

Tab này chưa kết nối PMW3901, chỉ test khả năng generate random point của EPS32 lên màn hình

ARDUINO CODE

```
#include <SPI.h>

#define FRAME_WIDTH 35
#define FRAME_HEIGHT 35
#define TOTAL_PIXELS (FRAME_WIDTH * FRAME_HEIGHT) // 1225 pixels

uint8_t frameBuffer[TOTAL_PIXELS];

void setup() {
  Serial.begin(115200);

  // LƯU Ý: Tuyệt đối không dùng vòng lặp while(!Serial) ở đây đối với ESP32-S3

  // TODO: Khởi tạo SPI và cấu hình chip PMW3901 của bạn ở đây...
}

void loop() {
  // --- 1. ĐỌC DỮ LIỆU TỪ CẢM BIẾN ---
  // Hiện tại đang tạo dữ liệu ngẫu nhiên (nhiều trắng) để test.
  // Khi lắp cảm biến, bạn thay vòng for này bằng hàm đọc SPI từ PMW3901.
  for(int i = 0; i < TOTAL_PIXELS; i++) {
    frameBuffer[i] = random(0, 256);
  }

  // --- 2. GỬI DỮ LIỆU LÊN MÁY TÍNH ---
  Serial.write('*'); // Bắn 1 byte mang ký tự '*' để làm cờ báo hiệu bắt đầu ảnh

  Serial.write(frameBuffer, TOTAL_PIXELS); // Bắn 1 loạt 1225 byte đi với tốc độ cao

  // Giới hạn tốc độ gửi (50ms ~ 20 khung hình/giây) để máy tính xử lý kịp
  delay(50);
}
```

PROCESSING CODE

```
import processing.serial.*;

Serial myPort;
int frameWidth = 35;
int frameHeight = 35;
int scaleFactor = 15; // Phóng to mỗi pixel lên 15 lần cho dễ nhìn (kích thước 525x525)
```

```

byte[] frameBuffer = new byte[1225];
int byteCount = 0;
boolean isReadingFrame = false;

void setup() {
  size(525, 525);
  noStroke(); // Tắt viền đen của các điểm ảnh

  // Mở cổng COM18 giao tiếp với ESP32
  myPort = new Serial(this, "COM18", 115200);
}

void draw() {
  // Liên tục hút dữ liệu từ cổng COM18
  while (myPort.available() > 0) {
    int inByte = myPort.read();

    if (!isReadingFrame) {
      // Đang ở trạng thái 1: TÌM CỜ BÁO HIỆU
      if (inByte == '*') {
        isReadingFrame = true; // Tìm thấy cờ -> Chuyển trạng thái
        byteCount = 0;        // Reset bộ đếm để chuẩn bị hứng 1225 byte
      }
    } else {
      // Đang ở trạng thái 2: HỨNG DỮ LIỆU TỪNG PIXEL
      frameBuffer[byteCount] = (byte)inByte;
      byteCount++;

      // Nếu đã hứng đủ 1225 pixel
      if (byteCount == frameWidth * frameHeight) {
        drawFrame(); // Gọi hàm vẽ ngay lập tức!
        isReadingFrame = false; // Reset lại trạng thái tìm cờ '*' cho khung hình tiếp theo
      }
    }
  }
}

// Hàm thực hiện việc vẽ ảnh ra màn hình
void drawFrame() {
  for (int y = 0; y < frameHeight; y++) {
    for (int x = 0; x < frameWidth; x++) {
      int i = y * frameWidth + x;
      int pixelValue = frameBuffer[i] & 0xff; // Chuyển byte có dấu thành số nguyên 0-255

      fill(pixelValue); // Đặt màu xám tương ứng với độ sáng của pixel
      rect(x * scaleFactor, y * scaleFactor, scaleFactor, scaleFactor); // Vẽ hình vuông
    }
  }
}

```

RESULTS

